

Embedded System Course 2022-23

C4µC

C/C++ Programming for Microcontrollers

Lecturer: Ing. Marco Santic

marco.santic@univaq.it

C4μC



- Introduction
 - Atmega328P
 - Toolchain
 - IDE
 - Examples
 - **C language recap**
- Basic functionalities
 - GPIO
 - Timers
 - Interrupt
 - Examples
- Comm. Interfaces
 - I2C
 - UART
 - SPI
 - Examples
- Other peripherals
 - ADC
 - GPIO bitbanging
 - Examples

C4μC - Examples

- Maximum value in an array (`1_max.c`)
- Simplest code



C4μC - Examples

- Maximum value in an array (`1_max.c`)
- Simplest code

```
#define A_SIZE 10
uint8_t array[A_SIZE] = {1,2,3,4,5,6,7,8,9,10};

int main(void)
{
    int i;
    uint8_t max = 0;

    for (i=0; i<A_SIZE; i++) {
        if (array[i] > max)
            max = array[i];
    }
    return max;
}
```

- Copy/paste in sketch `recap_01_max`, then compile and upload

C4μC - Examples

- Maximum value in an array (`1_max.c`)
- Comments:
 - The code compiles
 - It probably works, but we are missing the result!!!
 - What can we use to get the result?



C4μC - Examples

- Maximum value in an array (`2_max.c`)
- Serial interface for output



C4μC - Examples

- Maximum value in an array (`2_max.c`)
- Serial interface for output

```
#define A_SIZE 10
uint8_t array[A_SIZE] = {5,7,8,1,2,10,3,4,6,9};

void setup() {
    Serial.begin(9600);
    while (!Serial);
}

void loop() {
    int i;
    uint8_t max = 0;

    for (i=0; i<A_SIZE; i++) {
        if (array[i] > max) {
            max = array[i];
            Serial.print("Current max value: ");
            Serial.print(max);
            Serial.print('\n');
        }
    }

    Serial.print("\nFinal max value: ");
    Serial.print(max);
    Serial.print("\n\n");
    delay(5000);
}
```

- Copy/paste in sketch `recap_02_max`, then compile and upload

C4μC - Examples

- Maximum value in an array (`2_max.c`)
- Comments
 - We are using the serial communication port to get the result
 - We are using the IDE functions `setup()` and `loop()`



C4μC - Examples

- Maximum value in an array (`z_max.c`)
- Reading an external input via Serial interface



C4μC - Examples

- Maximum value in an array (`3_max.c`)
- Reading an external input via Serial interface

```
#define A_SIZE 10
uint8_t array[A_SIZE] = {5,7,8,1,2,10,3,4,6,9};
void setup() { Serial.begin(9600); while (!Serial);}

void loop() {
    int i;
    uint8_t max = 0;
    for (i=6; i<A_SIZE; i++) {
        Serial.print("Insert a value: ");
        while (!Serial.available());
        Serial.write(array[i] = Serial.read());
        Serial.print('\n');
    }
    for (i=0; i<A_SIZE; i++) {
        if (array[i] > max) {
            max = array[i];
            Serial.print("Current max value: ");
            Serial.print(max);
            Serial.print('\n');
        }
    }
    Serial.print("\nFinal max value: ");
    Serial.print(max);
    Serial.print("\n\n");
    delay(5000);
}
```

- Copy/paste in sketch `recap_03_max`, then compile and upload

C4μC - Examples



- Maximum value in an array (`3_max.c`)

- Comments

- We can send numbers from 0 to 9

- ... overwriting the array from the 7th position

idx	0	1	2	3	4	5	6	7	8	9
val	5	7	8	1	2	10	*	*	*	*

- How it is possible than the max is not 10?
 - Which values have we really inserted?
- ASCII Table...

C4μC - Examples



- Maximum value in an array (`z_max.c`)

ASCII TABLE

Decimal	Hex	Char	Decimal	Hex	Char	Decimal	Hex	Char	Decimal	Hex	Char
0	0	[NULL]	32	20	[SPACE]	64	40	@	96	60	`
1	1	[START OF HEADING]	33	21	!	65	41	A	97	61	a
2	2	[START OF TEXT]	34	22	"	66	42	B	98	62	b
3	3	[END OF TEXT]	35	23	#	67	43	C	99	63	c
4	4	[END OF TRANSMISSION]	36	24	\$	68	44	D	100	64	d
5	5	[ENQUIRY]	37	25	%	69	45	E	101	65	e
6	6	[ACKNOWLEDGE]	38	26	&	70	46	F	102	66	f
7	7	[BELL]	39	27	'	71	47	G	103	67	g
8	8	[BACKSPACE]	40	28	(72	48	H	104	68	h
9	9	[HORIZONTAL TAB]	41	29)	73	49	I	105	69	i
10	A	[LINE FEED]	42	2A	*	74	4A	J	106	6A	j
11	B	[VERTICAL TAB]	43	2B	+	75	4B	K	107	6B	k
12	C	[FORM FEED]	44	2C	,	76	4C	L	108	6C	l
13	D	[CARRIAGE RETURN]	45	2D	.	77	4D	M	109	6D	m
14	E	[SHIFT OUT]	46	2E	.	78	4E	N	110	6E	n
15	F	[SHIFT IN]	47	2F	/	79	4F	O	111	6F	o
16	10	[DATA LINK ESCAPE]	48	30	0	80	50	P	112	70	p
17	11	[DEVICE CONTROL 1]	49	31	1	81	51	Q	113	71	q
18	12	[DEVICE CONTROL 2]	50	32	2	82	52	R	114	72	r
19	13	[DEVICE CONTROL 3]	51	33	3	83	53	S	115	73	s
20	14	[DEVICE CONTROL 4]	52	34	4	84	54	T	116	74	t
21	15	[NEGATIVE ACKNOWLEDGE]	53	35	5	85	55	U	117	75	u
22	16	[SYNCHRONOUS IDLE]	54	36	6	86	56	V	118	76	v
23	17	[END OF TRANS. BLOCK]	55	37	7	87	57	W	119	77	w
24	18	[CANCEL]	56	38	8	88	58	X	120	78	x
25	19	[END OF MEDIUM]	57	39	9	89	59	Y	121	79	y
26	1A	[SUBSTITUTE]	58	3A	:	90	5A	Z	122	7A	z
27	1B	[ESCAPE]	59	3B	;	91	5B	[123	7B	{
28	1C	[FILE SEPARATOR]	60	3C	<	92	5C	\	124	7C	
29	1D	[GROUP SEPARATOR]	61	3D	=	93	5D]	125	7D	}
30	1E	[RECORD SEPARATOR]	62	3E	>	94	5E	^	126	7E	~
31	1F	[UNIT SEPARATOR]	63	3F	?	95	5F	-	127	7F	[DEL]

C4μC - Examples



- Maximum value in an array (`4_max.c`)
- `Serial.parseInt()`

```
#define A_SIZE 10
uint8_t array[A_SIZE] = {5,7,8,1,2,10,3,4,6,9};
void setup() { Serial.begin(9600); while (!Serial);}

void loop() {
    int i;
    uint8_t max = 0;
    for (i=6; i<A_SIZE; i++) {
        Serial.print("Insert a value: ");
        while (!Serial.available());
        array[i] = Serial.parseInt();
        Serial.print(array[i]);
        Serial.print('\n');
    }
    for (i=0; i<A_SIZE; i++) {
        if (array[i] > max) {
            max = array[i];
            Serial.print("Current max value: ");
            Serial.print(max);
            Serial.print('\n');
        }
    }
    Serial.print("\nFinal max value: ");
    Serial.print(max);
    Serial.print("\n\n");
    delay(5000);
}
```

- Copy/paste in sketch `recap_04_max`, then compile and upload

C4μC - Examples

- Maximum value in an array (`4_max.c`)
- Comments
 - With `Serial.parseInt()` we are now interpreting the characters and generating integers
 - What if we insert a value greater than 255?
and if we insert a negative value?
 - Note the *variable types* and the *casts or truncations*



C4μC - Examples

- Be aware of what you read and write (`5_rdwr.c`)

```
#define A_SIZE 10
int8_t array[A_SIZE] = {0,0,0,0,0,0,0,0,0,0};

void setup() {
    Serial.begin(9600);
    while (!Serial);
}

void loop() {
    int i;
    for (i=0; i<A_SIZE; i++) {
        Serial.print("Insert a value: ");
        while (!Serial.available());
        array[i] = Serial.parseInt();
        Serial.print(array[i]);
        Serial.print('\n');
    }

    Serial.print(".print .write");
    for (i=0; i<A_SIZE; i++) {
        Serial.print(" ");
        Serial.print(array[i]);
        Serial.print(" ");
        Serial.write(array[i]);
        Serial.print('\n');
    }
    delay(5000);
}
```

- Copy/paste in sketch `recap_05_max`, then compile and upload

C4μC - Examples

- Be aware of what you read and write (`5_rdwr.c`)
- Comments
 - A number can be truncated: `uint8_t = (int)...`
 - Wrong sign interpretation: `uint_x -> int_x` (ex: 255 --> -1)
 - During reading and writing
 - ...

C4μC - Examples

- Mean value in an array (`6_avg.c`)
- Read 10 values and compute the mean value



C4μC - Examples

- Mean value in an array (`6_avg.c`)

```
#define A_SIZE 10
uint8_t array[A_SIZE];

void setup() {
    Serial.begin(9600); while (!Serial);}

void loop() {
    int i;
    uint8_t avg = 0;

    for (i=0; i<A_SIZE; i++) {
        Serial.print("Insert a value: ");
        while (!Serial.available());
        array[i] = Serial.parseInt();
        Serial.print(array[i]);
        Serial.print('\n');
    }

    for (i=0; i<A_SIZE; i++) {
        avg += array[i];
    }

    avg /= 10; // ERROR?!
    Serial.print("\nAverage value: ");
    Serial.print(avg);
    Serial.print("\n");
    delay(5000);
}
```

ERROR?!

- Copy/paste in sketch `recap_06_avg`, then compile and upload

C4μC - Examples

- Mean value in an sliding windows (`7_avg.c`)
- What if we have to compute the mean of the last n values that we receive in a continuous stream?



C4μC - Examples

- Mean value in an sliding windows (`7_avg.c`)

```
#define WIN_SIZE 8
uint8_t window[WIN_SIZE];
int i = 0;
int avg = 0;

void setup() { Serial.begin(9600); while (!Serial);
               Serial.println("Window size: 8");
}

void loop() {
    int j;
    while (!Serial.available());
    window[i % WIN_SIZE] = Serial.parseInt();
    Serial.print((int)window[i]);
    Serial.print('\n');
    i++;

    avg = 0;
    for (j=0; j<WIN_SIZE; j++) {
        Serial.print(window[j]);
        Serial.print(' ');
        avg += window[j];
    }
    Serial.print('\n');

    avg /= WIN_SIZE;
    Serial.print("\nAverage = ");
    Serial.print(avg);
    Serial.print('\n');
}
```

- Copy/paste in sketch `recap_07_avg`, then compile and upload

C4μC - Examples



- Mean value in an sliding windows (`8_avg.c`)
- Comments
 - On every value read, we compute the mean for the whole array
- Can we "improve a little" the performances?
 - Imagine the window is very large... (1000 values)
 - ...let's avoid to make all the sums

C4μC - Examples

- Mean value in an sliding windows (`8_avg.c`)

```
#define WIN_SIZE 8
uint8_t window[WIN_SIZE] = {0,0,0,0,0,0,0,0};
int i = 0;
int sum = 0;
int last = 0;
int avg = 0;

void setup() {
    Serial.begin(9600);
    while (!Serial);
    Serial.println("Window size: 8");
}

void loop() {
    while (!Serial.available());
    last = window[ i % WIN_SIZE];
    window[i % WIN_SIZE] = Serial.parseInt();
    sum += window[i % WIN_SIZE];
    sum -= last;
    Serial.print((unsigned int)window[i % WIN_SIZE]);
    Serial.print('\n');
    i++;
    avg = sum / WIN_SIZE;
    Serial.print("\nAverage = ");
    Serial.print(avg);
    Serial.print('\n');
}
```

- Copy/paste in sketch `recap_08_avg`, then compile and upload

C4μC - Examples

- Identification of values in a stream (`9_seq.c`)
- Try to get only characters that represent a number [0-9] and put them in an array



C4μC - Examples

- Identification of values in a stream (`9_seq.c`)

```
#define A_SIZE 10
uint8_t array[A_SIZE];

void setup() {
    Serial.begin(9600);
    while (!Serial);
}

void loop() {
    int i = 0;
    int j = 0;
    uint8_t c;

    for (i=0; i<A_SIZE; i++) {
        while (!Serial.available());
        c = Serial.read();
        if (c < '0' || c > '9')
            break;
        array[i] = c; /* ... = c - '0'; */
        Serial.print((unsigned int)array[i]);
        Serial.print('\n');
    }

    for (j=0; j<i; j++) {
        Serial.print(array[j]);
        Serial.print(' ');
    }
    delay(5000);
}
```

- Copy/paste in sketch `recap_09_seq`, then compile and upload

C4μC - Examples

- Homemade `parseInt` (`10_seq.c`)
- Starting from the previous example
 - Read only chars [0-9] and put them in the array
 - Then convert the inserted values in an integer
- How would you do it?



C4μC - Examples



- Homemade parseInt (`10_seq.c`)

```
#define A_SIZE 10
uint8_t array[A_SIZE];
uint16_t v;

void setup() {
    Serial.begin(9600);
    while (!Serial);
}

void loop() {
    int i = 0;
    int j = 0;
    int k = 1;
    uint8_t c;

    for (i=0; i<A_SIZE; i++) {
        while (!Serial.available());
        c = Serial.read();
        if (c < '0' || c > '9')
            break;
        array[i] = c - 0x30;
        Serial.print((unsigned int)array[i]);
        Serial.print('\n');
    }
}

//... loop() continues
for (j=0; j<i; j++) {
    Serial.print(array[j]);
    Serial.print(' ');
}

i--;
v = 0;


while (i >= 0) {
    v += array[i] * k;
    k *= 10;
    i--;
}

Serial.print("\nNumber = ");
Serial.print(v);
Serial.print('\n');

delay(5000);
```

- Copy/paste in sketch `recap_10_seq`, then compile and upload

C4μC - Examples



Any question?